[3.13] Find matrix minor and adjoint

The 89/92+ have no built-in functions to find the minor and adjoint of a matrix, but these are easily accomplished with the built-in functions. The minor of an n x n square matrix A = $[a_{ij}]$ is the determinant of the matrix that results from deleting row *i* and column *j* of A. If the minor is

$$\left| M_{ij} \right|$$

then the cofactor of a_{ij} is defined as

$$(-1)^{i+j}\left| M_{ij} \right| = \alpha_{ij}$$

and the adjoint of A is defined as

$$\mathrm{adj}(A) = \begin{bmatrix} \alpha_{11} & \alpha_{21} & \cdots & \alpha_{n1} \\ \alpha_{12} & \alpha_{22} & \cdots & \alpha_{n2} \\ \cdots & \cdots & \cdots & \cdots \\ \alpha_{1n} & \alpha_{2n} & \cdots & \alpha_{nn} \end{bmatrix} \qquad [1]$$

However, we need not find the minors and cofactors to find the adjoint, because of this identity:

$$\mathrm{adj}(A) = A^{-1} \cdot |A| \ \text{ if } |A| \neq 0 \qquad [2]$$

It is faster to calculate the adjoint with this identity than by finding the minors and cofactors, but the identity is true only if the matrix is non-singular. However, a singular matrix also has an adjoint. So, an optimized adjoint function uses [1] if the matrix is singular, and [2] if not. This function finds the adjoint of a matrix:

```
adjoint(m)
Func
©(m) Return adjoint of square matrix m
local k,n,i,j,d

det(m)→d

if d≠0 or gettype(d)="EXPR" then

 m^(-1)*d→n

else
 rowdim(m)→k
 newmat(k,k)→n

 for i,1,k
  for j,1,k
   (-1)^(i+j)*det((mrowdel(mrowdel(m,i)ᵀ,j))ᵀ)→n[j,i]
  endfor
 endfor

endif

return n

EndFunc
```

adjoint() uses equation [2] if the matrix is non-singular, or if the matrix is symbolic. There is no error checking, and a *Dimension* error occurs if the matrix is not square.

The process of finding the matrix minor is built into *adjoint(),* but *mminor()* returns the minor, if needed:

```
mminor(m,r,c)
Func
©(m,r,c) Return first minor r,c of matrix m

return det((mrowdel(mrowdel(m,r)ᵀ,c))ᵀ)

EndFunc
```

adjoint() and *mminor()* call *mrowdel(),* which is described in tip [3.3].

To find the adjoint of $\begin{bmatrix} a & b \\ c & d \end{bmatrix}$

use this call:

```
adjoint([[a,b][c,d]]
```

which returns $\begin{bmatrix} d & -b \\ -c & a \end{bmatrix}$

To find the adjoint of $\begin{bmatrix} 1 & 2 & 3 \\ 1 & 3 & 4 \\ 1 & 4 & 3 \end{bmatrix}$

use this call

```
adjoint([[1,2,3][1,3,4][1,4,3]])
```

which returns $\begin{bmatrix} -7 & 6 & -1 \\ 1 & 0 & -1 \\ 1 & -2 & 1 \end{bmatrix}$

As an example of a singular matrix, find the adjoint of $\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & 2 \\ 0 & 0 & 0 \end{bmatrix}$

with

```
adjoint([[1,2,3][0,1,2][0,0,0]])
```

which returns $\begin{bmatrix} 0 & 0 & 1 \\ 0 & 0 & -2 \\ 0 & 0 & 1 \end{bmatrix}$

(Credit to Mike Roberts for pointing out equation [2])